
1. [bookmark: _Toc122434485][bookmark: _Toc126242285]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D23wk37 related to the title of this CR.
Contact:	Shaun Harry 
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc126242286]Corrections required
1.1 Change 1

	Function name
	fl_TC_11_3_9_TestBody()

	Reason for change
	For test case 11.3.9a, the power level change for T0 should be made after the UE has been switched on to apply SNPN-related AT commands, otherwise the UE may camp unexpectedly. 

	Summary of change
	Move power level change for T0

	TTCN module
	UAC_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	  
function fl_TC_11_3_9and9a_NR5GC_Common(NR_SysinfoCombination_Type p_NR_SysinfoCombination) runs on NR5GC_PTC
   {
    const octetstring tsc_Criteria_DNN := '00'O;  //DNN Type  // @sic R5s210307 sic@
    var template (value) NR_CellPowerList_Type v_CellPowerList_T0;
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var octetstring v_OctetString;
    var octetstring v_Criteria;
    
    v_OctetString := int2oct(lengthof(tsc_DNNValue),1) & char2oct(tsc_DNNValue); // @sic R5s210307 sic@
    
    v_Criteria := tsc_Criteria_DNN & '01'O & int2oct(lengthof(v_OctetString),1) & v_OctetString; // @sic R5s210307 sic@

    
    // System information combination NR-12 as defined in TS 38.508-1 clause 4.4.3.1.2 is used in NR cells.
    f_NR5GC_Init(p_NR_SysinfoCombination);
    
    // Table 11.3.9.3.3-2: SIB1 for NR Cell 1 and NR Cell 12
    // @sic R5s210307 sic@
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1,
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(33,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1111111'B)));
     
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell12, // @sic R5-220559 sic@
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(33,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1111111'B)));

    //Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);
    f_NR_CellConfig_Def(nr_Cell12);
    
    v_CellPowerList_T0 := {//@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2),
      cs_NR_CellPower (nr_Cell12, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)
    };
    
    f_NR_SetCellPowerList(v_CellPowerList_T0);
    
    if( f_GetTestcaseAttrib_NR_SNPN(testcasename()))
    {
      f_UT_SwitchOnUE(UT);
      f_UT_ProvideSubscriberDataList(UT, f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN(nr_Cell1)), 0);
      f_UT_ProvideSubscriberDataList(UT, f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN(nr_Cell12)), 1);
      f_UT_SwitchOffUE(UT, false);

    }
     
    // The UE is registered on PLMN1 (NR Cell 1) and in state 3N-A on NR Cell 1(serving cell) by using the procedure described in TS 38.508-1 [4] clause 4.5.2.2 with 'connected without release'
    // except that the REGISTRATION ACCEPT message indicates Operator Defined Access Category ""33"" as described in Table 11.3.9.3.3-1.
    v_GMM_MobilityInfo := f_NR5GC_Preamble_Steps1_13(nr_Cell1);
    
    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,
                                                   v_GMM_MobilityInfo,  // @sic R5-206426 sic@
                                                   f_NR5GC_GetMobileIdGUTI (nr_Cell1),
                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell1)), {bit2oct(f_NR_CellInfo_GetTAC (nr_Cell1))}),
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   cs_OperatorAccessCatDefinitions(cs_AccessCatDefinition ('00'O,'1'B,'00001'B, v_Criteria)));
    
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell1,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
                                   
    // Steps 15-19a1 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed.
    f_NR5GC_RRC_Idle_Steps15_20(nr_Cell1,TESTMode_OFF);
    
    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1);
    
    f_NR_TestBody_Set(true);

    f_NR_TestBody_Set(true);// @sic R5s210307 sic@
    fl_TC_11_3_9_TestBody();
    f_NR_TestBody_Set(false);
    
    f_NR_Postamble(nr_Cell12, STATE_CONNECTED_3A);// @sic R5s210307 sic@
  
  }








After Change:
	function fl_TC_11_3_9and9a_NR5GC_Common(NR_SysinfoCombination_Type p_NR_SysinfoCombination) runs on NR5GC_PTC
   {
    const octetstring tsc_Criteria_DNN := '00'O;  //DNN Type  // @sic R5s210307 sic@
    var template (value) NR_CellPowerList_Type v_CellPowerList_T0;
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var octetstring v_OctetString;
    var octetstring v_Criteria;
    
    v_OctetString := int2oct(lengthof(tsc_DNNValue),1) & char2oct(tsc_DNNValue); // @sic R5s210307 sic@
    
    v_Criteria := tsc_Criteria_DNN & '01'O & int2oct(lengthof(v_OctetString),1) & v_OctetString; // @sic R5s210307 sic@

    
    // System information combination NR-12 as defined in TS 38.508-1 clause 4.4.3.1.2 is used in NR cells.
    f_NR5GC_Init(p_NR_SysinfoCombination);
    
    // Table 11.3.9.3.3-2: SIB1 for NR Cell 1 and NR Cell 12
    // @sic R5s210307 sic@
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1,
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(33,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1111111'B)));
     
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell12, // @sic R5-220559 sic@
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(33,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1111111'B)));

    //Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);
    f_NR_CellConfig_Def(nr_Cell12);
    
    v_CellPowerList_T0 := {//@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2),
      cs_NR_CellPower (nr_Cell12, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)
    };
    
    //MOVED f_NR_SetCellPowerList(v_CellPowerList_T0);
    
    if( f_GetTestcaseAttrib_NR_SNPN(testcasename()))
    {
      f_UT_SwitchOnUE(UT);
      f_UT_ProvideSubscriberDataList(UT, f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN(nr_Cell1)), 0);
      f_UT_ProvideSubscriberDataList(UT, f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN(nr_Cell12)), 1);
      f_UT_SwitchOffUE(UT, false);

    }

     f_NR_SetCellPowerList(v_CellPowerList_T0);
     
    // The UE is registered on PLMN1 (NR Cell 1) and in state 3N-A on NR Cell 1(serving cell) by using the procedure described in TS 38.508-1 [4] clause 4.5.2.2 with 'connected without release'
    // except that the REGISTRATION ACCEPT message indicates Operator Defined Access Category ""33"" as described in Table 11.3.9.3.3-1.
    v_GMM_MobilityInfo := f_NR5GC_Preamble_Steps1_13(nr_Cell1);
    
    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,
                                                   v_GMM_MobilityInfo,  // @sic R5-206426 sic@
                                                   f_NR5GC_GetMobileIdGUTI (nr_Cell1),
                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell1)), {bit2oct(f_NR_CellInfo_GetTAC (nr_Cell1))}),
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   cs_OperatorAccessCatDefinitions(cs_AccessCatDefinition ('00'O,'1'B,'00001'B, v_Criteria)));
    
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell1,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
                                   
    // Steps 15-19a1 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed.
    f_NR5GC_RRC_Idle_Steps15_20(nr_Cell1,TESTMode_OFF);
    
    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1);
    
    f_NR_TestBody_Set(true);

    f_NR_TestBody_Set(true);// @sic R5s210307 sic@
    fl_TC_11_3_9_TestBody();
    f_NR_TestBody_Set(false);
    
    f_NR_Postamble(nr_Cell12, STATE_CONNECTED_3A);// @sic R5s210307 sic@
  
  }






1.2 [bookmark: _Toc139301353]Change 2

	Function name
	fl_TC_11_3_9_TestBody()

	Reason for change
	[bookmark: _Toc131396982]According to 24.501 Annex C.2, the UE NAS Security Context is stored on a per-SNPN basis, so a UE will delete this when selecting a different SNPN and so will have no existing security context for the initial registration procedure at steps 4-22a1 in the case of test case 11.3.9a.

C.2      Storage of 5GMM information for UEs operating in SNPN access operation mode
The 5GMM information for UEs operating in SNPN access operation mode and not registering or registered for the onboarding service in SNPN are stored according to the following conditions:
-     if the UE does not support access to an SNPN using credentials from a credentials holder and equivalent SNPNs, the following 5GMM parameters shall be stored per subscribed SNPN in a non-volatile memory in the ME together with the subscriber identifier associated with the SNPN identity of the SNPN in the "list of subscriber data" configured in the ME (see 3GPP TS 23.122 [5]) or with the SUPI from the USIM if no subscriber identifier is configured in the entry of the "list of subscriber data" associated with the SNPN identity and the UE has a valid USIM;and
……………
d)   5G NAS security context parameters from a full native 5G NAS security context (see 3GPP TS 33.501 [24]);
………………



	Summary of change
	Use condition Initial_NoSecurity

Note : An associated prose CR on 38.523-1 will be raised at RAN5#101

	TTCN module
	UAC_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	  function fl_TC_11_3_9_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_RadioBearerId_Type v_SRB:= tsc_NR_RbId_SRB2;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T1;
    var template (omit) DNN v_DNN;
    var boolean v_ReceiveRRCSetupRequest;
    var default v_MyDefaultVar:= null; // @sic R5-224717 sic@
    timer t_Wait := 11.2; // @sic R5-213507 sic@
    timer t_WaitTime := 5.0; // @sic R5-214783 sic@

    v_DNN := cs_DNN(-, f_DomainName_Encode(tsc_DNNValue)); // @sic R5s210307 sic@
    
    //@siclog "Step 3" siclog@
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
    
   //@siclog "Step 4" siclog@
   // Check: Does the UE transmit a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is the same value in UL NAS TRANSPORT message in preamble within 20s?
    t_Wait.start;
    alt{
        []SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
            {
               // @sic R5s210307 sic@
               f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
            }
        []t_Wait.timeout
            {
               f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
            }
    }
    // @siclog "Step 5" siclog@
    //The SS adjusts the NR Cells power levels according to row "T1" in table 11.3.9.3.2-1/2.
     v_CellPowerList_T1 := { //@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell12, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)
    };
    
    f_NR_SetCellPowerList(v_CellPowerList_T1);
    
    if( f_GetTestcaseAttrib_NR_SNPN(testcasename())) //@sic R5-233351 sic@
    {
     //@siclog "Step 4-22a1" siclog@
     //Steps 2-20a1 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed
      f_NR5GC_MobileInfo_Init(); // Need to 'reset' all stored lists as all DRBs and PDU sessions do not exist in the new SNPN
      f_NR5GC_RRC_IdleState1N_Def (nr_Cell12);
    } else {
     // @siclog "Step 6" siclog@
     //The mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 is performed on NR Cell 12 and release RRC connection (Note 1)
     //Note 1:   SS ignore the PDU SESSION ESTABLISHMENT REQUEST sent before SS release RRC connection. @sic R5-224717 sic@
      v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(nr_Cell12));
      f_NR5GC_MobilityRegistration(nr_Cell12, -, -, -, -, noRrcConnectionRelease);
      f_NR_RRCRelease(nr_Cell12);
      deactivate(v_MyDefaultVar);
    }
 
    // @sic R5-214783 sic@
    // @siclog "Steps 6A - 6Bb1" siclog@
    t_WaitTime.start;
    alt{
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_MoData, ?)))
            {
              t_WaitTime.stop;
              f_NR_ULGrantConfiguration_Start(nr_Cell12, -, -, -, cs_NR_UplinkTimeAlignment_Stop, -, -, tsc_NoCnfReq ); // @sic R5s221054 sic@
              f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6Ba1");
            
            }
        [] t_WaitTime.timeout
            {
              //@siclog "Step 6Bb2" siclog@
              // Make the UE attempt to establish a PDU session for the DNN defined in Table 11.3.9.3.3-1. This can be done by an AT/MMI command.
                f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
                //@siclog "Step 6Bb3" siclog@
                //Check: Does the UE transmit a RRCSetupRequest message and establishmentCause is set to mo-Data?
                v_ReceiveRRCSetupRequest := f_NR_RRC_SetupRequest_Common( nr_Cell12, t_Wait, cr_NR_EstablishmentCause_MoData);
                if (v_ReceiveRRCSetupRequest) {
                    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6Bb3");
                } else {
                    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6Bb3");
                }
                
            }
    }
    
    //@siclog "Steps 9 - 14" siclog@
    //Steps 3-8 of the generic procedure for NR RRC_Connected specified in TS 38.508-1 Table 4.5.4.2-3 are performed.
    f_NR_RRCSetup_Def (nr_Cell12);
    
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *),
                                          tsc_SHT_IntegrityProtected);
                                          
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)),'0000'B)) {  // @sic R5s200654 R5s210307 sic@
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg);

    //@siclog "Steps 15 - 16" siclog@
    //The UE transmits a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is set to the same value with the DNN value of Operator-defined access category definitions defined in Table 11.3.9.3.3-1
    //The SS transmits PDU SESSION ESTABLISHMENT ACCEPT message.
    f_NR5GC_PDUSessionEstablishment (nr_Cell12, 1, -, -, - , v_DNN, -, v_DNN);
    
  }









After Change:
	function fl_TC_11_3_9_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_RadioBearerId_Type v_SRB:= tsc_NR_RbId_SRB2;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T1;
    var template (omit) DNN v_DNN;
    var boolean v_ReceiveRRCSetupRequest;
    var default v_MyDefaultVar:= null; // @sic R5-224717 sic@
    timer t_Wait := 11.2; // @sic R5-213507 sic@
    timer t_WaitTime := 5.0; // @sic R5-214783 sic@

    v_DNN := cs_DNN(-, f_DomainName_Encode(tsc_DNNValue)); // @sic R5s210307 sic@
    
    //@siclog "Step 3" siclog@
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
    
   //@siclog "Step 4" siclog@
   // Check: Does the UE transmit a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is the same value in UL NAS TRANSPORT message in preamble within 20s?
    t_Wait.start;
    alt{
        []SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
            {
               // @sic R5s210307 sic@
               f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
            }
        []t_Wait.timeout
            {
               f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
            }
    }
    // @siclog "Step 5" siclog@
    //The SS adjusts the NR Cells power levels according to row "T1" in table 11.3.9.3.2-1/2.
     v_CellPowerList_T1 := { //@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell12, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)
    };
    
    f_NR_SetCellPowerList(v_CellPowerList_T1);
    
    if( f_GetTestcaseAttrib_NR_SNPN(testcasename())) //@sic R5-233351 sic@
    {
     //@siclog "Step 4-22a1" siclog@
     //Steps 2-20a1 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed
      f_NR5GC_MobileInfo_Init(); // Need to 'reset' all stored lists as all DRBs and PDU sessions do not exist in the new SNPN
      f_NR5GC_RRC_IdleState1N_Def (nr_Cell12,-,-, Initial_NoSecurity);
    } else {
     // @siclog "Step 6" siclog@
     //The mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 is performed on NR Cell 12 and release RRC connection (Note 1)
     //Note 1:   SS ignore the PDU SESSION ESTABLISHMENT REQUEST sent before SS release RRC connection. @sic R5-224717 sic@
      v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(nr_Cell12));
      f_NR5GC_MobilityRegistration(nr_Cell12, -, -, -, -, noRrcConnectionRelease);
      f_NR_RRCRelease(nr_Cell12);
      deactivate(v_MyDefaultVar);
    }
 
    // @sic R5-214783 sic@
    // @siclog "Steps 6A - 6Bb1" siclog@
    t_WaitTime.start;
    alt{
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_MoData, ?)))
            {
              t_WaitTime.stop;
              f_NR_ULGrantConfiguration_Start(nr_Cell12, -, -, -, cs_NR_UplinkTimeAlignment_Stop, -, -, tsc_NoCnfReq ); // @sic R5s221054 sic@
              f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6Ba1");
            
            }
        [] t_WaitTime.timeout
            {
              //@siclog "Step 6Bb2" siclog@
              // Make the UE attempt to establish a PDU session for the DNN defined in Table 11.3.9.3.3-1. This can be done by an AT/MMI command.
                f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
                //@siclog "Step 6Bb3" siclog@
                //Check: Does the UE transmit a RRCSetupRequest message and establishmentCause is set to mo-Data?
                v_ReceiveRRCSetupRequest := f_NR_RRC_SetupRequest_Common( nr_Cell12, t_Wait, cr_NR_EstablishmentCause_MoData);
                if (v_ReceiveRRCSetupRequest) {
                    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6Bb3");
                } else {
                    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6Bb3");
                }
                
            }
    }
    
    //@siclog "Steps 9 - 14" siclog@
    //Steps 3-8 of the generic procedure for NR RRC_Connected specified in TS 38.508-1 Table 4.5.4.2-3 are performed.
    f_NR_RRCSetup_Def (nr_Cell12);
    
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *),
                                          tsc_SHT_IntegrityProtected);
                                          
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)),'0000'B)) {  // @sic R5s200654 R5s210307 sic@
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg);

    //@siclog "Steps 15 - 16" siclog@
    //The UE transmits a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is set to the same value with the DNN value of Operator-defined access category definitions defined in Table 11.3.9.3.3-1
    //The SS transmits PDU SESSION ESTABLISHMENT ACCEPT message.
    f_NR5GC_PDUSessionEstablishment (nr_Cell12, 1, -, -, - , v_DNN, -, v_DNN);
    
  }





