
1. [bookmark: _Toc122434485][bookmark: _Toc126242285]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D23wk37 related to the title of this CR.
Contact:	Shaun Harry 
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc126242286]Corrections required

1. [bookmark: _Toc54888063][bookmark: _Toc100131355][bookmark: _Toc149040584][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 1
	Function name
	fl_TC_6_3_1_7_TestBody()

	Reason for change
	For the optional step 17Aa1 which takes place if 5GS registration type is set as Initial Registration in step 6, TTCN handling is incorrect. If the UE sends a 2nd PDU SESSION ESTABLISHMENT REQUEST after sending the PDU SESSION ESTABLISHMENT REQUEST with type “emergency”, this will be stored inside the default handler a_NR5GC_AdditionalPDUSessionEstablishmentReq but will never be processed due to p_ExpectedNumberOfNewPDUSessions being passed as 1 (from f_NR5GC_EmergencyCallPDUSession() to f_NR5GC_PDUSessionEstablishment()). Since the SS will never send a PDU SESSION ESTABLISHMENT ACCEPT for this request, the UE may re-send the PDU SESSION ESTABLISHMENT REQUEST message once the emergency call is released, causing test case failure.

	Summary of change
	Changes are proposed to handle both PDU Session Estalishment Requests for Emergency and IMS session in parallel.

Note : The solution is based on the MCC TF160 comments for R5s230685, and the function f_NR5GC_MsgInDefault_RemoveOneNASMessageReq() is defined in R5s230685_MCC160Comments_r1 .

	TTCN module
	SOR_NR5GC.ttcn

	MCC160 Comment
	


[bookmark: _Toc54888064]
Before Change:
	   
  function fl_TC_6_3_1_7_TestBody() runs on NR5GC_PTC
  {
    var NAS_PlmnId v_PlmnID_2 := f_NR_Asn2Nas_PlmnId(tsc_NR_HPLMN_002_11);    //@sic R5-204410 sic@
    var O2_Type v_AccessTechId :='0800'O;
    var octetstring v_Data := v_PlmnID_2 & v_AccessTechId;
    var template (value) SOR_Header v_Header := cs_NG_SOR_Header('1'B,'1'B,'1'B,'0'B);
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var default v_MyDefaultVar := null; //@sic R5-231404 sic@
    //@siclog "Step 1" siclog@
    //Power on the UE
    
    //@siclog "Step 1A1-1A13" siclog@
    //Steps 1 to 13 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed    //@sic R5-204410 sic@
     f_NR5GC_Preamble_Steps1_13(nr_Cell11);
     //@siclog "Step 1B" siclog@
    //SS transmits a REGISTRATION REJECT message with cause #22 (Congestion) and T3346 set to 3 minutes.    //@sic R5-204410 sic@
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_REGISTRATION_REJECT(cs_GMM_GSM_Cause(omit, tsc_NR5GCCause_Congestion), cs_GPRS_Timer2_3_IEI('5F'O, '001'B, '00011'B)))));
    //@siclog "Step 1C" siclog@
    //The SS releases the RRC connection    //@sic R5-204410 sic@
    f_NR_RRCRelease(nr_Cell11);
    //@siclog "Step 2" siclog@
    //The UE is made to establish an emergency call. This can be done by an AT/MMI command
    f_UT_RequestIMSEmergencyCall (UT);
    //@siclog "Step 3" siclog@
    // The UE transmits an RRCSetupRequest message on NR Cell 11 with establishmentCause set to 'emergency'.
    SRB.check(receive(car_NR_SRB0_RrcPdu_IND(nr_Cell11, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_emergency)))); //@sic R5s201466 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "RRCSetupRequest received"); //@sic R5s200305 sic@
    //@siclog "Steps 4-5" siclog@
    //Steps 3 to 4 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed
    //@siclog "Step 6" siclog@
    //The UE transmit a REGISTRATION REQUEST message with IE 5GS registration type set to "Emergency registration" or "Initial registration"//@sic R5-210807 R5s201466 sic@
    //@siclog "Steps 7-14" siclog@
    //Steps 5 to 13 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed on NR Cell 13
    //@siclog "Step 15" siclog@
    //The SS transmits a DLInformationTransfer message and a REGISTRATION ACCEPT message containing
    //steering of roaming information, includes unmatched SOR-MAC-IAUSF that will result in unsuccessful security check while on UE reception
    f_NR5GC_SorRegistration(nr_Cell11,v_Data,v_Header,true,Initial_Secure, tsc_SHT_IntegrityProtected);//@sic R5s210236 R5s201455 R5-205361 sic@//using Initial_Secure to indicate the security context state.
    //@siclog "Step 16" siclog@
    //The UE transmits an ULInformationTransfer message and REGISTRATION COMPLETE message without including an SOR transparent container
    f_NR5GC_Registration_Complete(nr_Cell11);
    
    //@siclog "Step 17" siclog@
    //Check: Does the UE transmit an UL NAS TRANSPORT message with "Request type" set to "initial emergency request", and a PDU SESSION ESTABLISHMENT REQUEST message on NR Cell 13
    //@siclog "Step 17Aa1" siclog@Exception
    // The generic procedure for UE-requested PDU session establishmentspecified in subclause 4.5A.2,
    //takes place performing establishment of UE-requested PDU session(s) with ExpectedNumberOfNewPDUSessions = pc_noOf_PDUsSameConnection.
    if (pc_noOf_PDUsSameConnection > 0 ){//@sic R5-227239 sic@
      SRB.check(receive(car_NR_SRB_NasPdu_IND(nr_Cell11, tsc_NR_RbId_SRB1, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp);
      if (f_Check_NG_ULNASTransport(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_PayloadContainerType('0001'B), omit,omit,cr_NG_Request_Type('001'B)))
        {
        f_NR5GC_PDUSessionEstablishment (nr_Cell11,-,tsc_NR_RbId_SRB1);
        }
      }
    
   
    //siclog "Step 18-22" siclog@//@sic R5-216165 sic@
    //Steps 9 to 13 of the procedure of "IMS Emergency call establishment in 5GC with IMS emergency registration" described in TS 38.508-1 [4] Table 4.9.11.2.2-1 are performed.
    f_NR5GC_EmergencyCallPDUSession(nr_Cell11,tsc_NR_RbId_SRB1);//@sic R5s211479 sic@ 
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2], "Emergency Call Established");  // Emergency call now active @sic R%s220982, R5-225362 sic@ 
    //@siclog "Step 23" siclog@//@sic R5-216165 sic@
    //Generic procedure of "Test Procedure for IMS MO Emergency call release" described in TS 38.508-1 [4] Table 4.9.12A.2.2-1 are performed.
    f_NR5GC_EmergencyCallRelease(nr_Cell11);//@sic R5s211479 sic@
    // The above function won't finish until IMS messages have been sent and PDU session has been released @sic R5s220982 sic@
    //@siclog "Step 24" siclog@//@sic R5-216165 sic@
    //The SS releases the RRC connection.
    v_MyDefaultVar := activate(a_NR5GC_Deregistration_REQ(nr_Cell11)); //@sic R5-231404 sic@
    f_NR_RRCRelease(nr_Cell11);
    deactivate(v_MyDefaultVar); //@sic R5-231404 sic@
    }  



After Change:
	   */
  function fl_TC_6_3_1_7_TestBody() runs on NR5GC_PTC
  {
    var NAS_PlmnId v_PlmnID_2 := f_NR_Asn2Nas_PlmnId(tsc_NR_HPLMN_002_11);    //@sic R5-204410 sic@
    var O2_Type v_AccessTechId :='0800'O;
    var octetstring v_Data := v_PlmnID_2 & v_AccessTechId;
    var template (value) SOR_Header v_Header := cs_NG_SOR_Header('1'B,'1'B,'1'B,'0'B);
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var default v_MyDefaultVar := null; //@sic R5-231404 sic@
    //@siclog "Step 1" siclog@
    //Power on the UE
    
    var NR_RadioBearerId_Type v_SRB := tsc_NR_RbId_SRB1;
    var boolean v_InitialRxd := false;
    var boolean v_EmgRxd := false;   

    //@siclog "Step 1A1-1A13" siclog@
    //Steps 1 to 13 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed    //@sic R5-204410 sic@
     f_NR5GC_Preamble_Steps1_13(nr_Cell11);
     //@siclog "Step 1B" siclog@
    //SS transmits a REGISTRATION REJECT message with cause #22 (Congestion) and T3346 set to 3 minutes.    //@sic R5-204410 sic@
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell11,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_REGISTRATION_REJECT(cs_GMM_GSM_Cause(omit, tsc_NR5GCCause_Congestion), cs_GPRS_Timer2_3_IEI('5F'O, '001'B, '00011'B)))));
    //@siclog "Step 1C" siclog@
    //The SS releases the RRC connection    //@sic R5-204410 sic@
    f_NR_RRCRelease(nr_Cell11);
    //@siclog "Step 2" siclog@
    //The UE is made to establish an emergency call. This can be done by an AT/MMI command
    f_UT_RequestIMSEmergencyCall (UT);
    //@siclog "Step 3" siclog@
    // The UE transmits an RRCSetupRequest message on NR Cell 11 with establishmentCause set to 'emergency'.
    SRB.check(receive(car_NR_SRB0_RrcPdu_IND(nr_Cell11, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_emergency)))); //@sic R5s201466 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "RRCSetupRequest received"); //@sic R5s200305 sic@
    //@siclog "Steps 4-5" siclog@
    //Steps 3 to 4 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed
    //@siclog "Step 6" siclog@
    //The UE transmit a REGISTRATION REQUEST message with IE 5GS registration type set to "Emergency registration" or "Initial registration"//@sic R5-210807 R5s201466 sic@
    //@siclog "Steps 7-14" siclog@
    //Steps 5 to 13 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed on NR Cell 13
    //@siclog "Step 15" siclog@
    //The SS transmits a DLInformationTransfer message and a REGISTRATION ACCEPT message containing
    //steering of roaming information, includes unmatched SOR-MAC-IAUSF that will result in unsuccessful security check while on UE reception
    f_NR5GC_SorRegistration(nr_Cell11,v_Data,v_Header,true,Initial_Secure, tsc_SHT_IntegrityProtected);//@sic R5s210236 R5s201455 R5-205361 sic@//using Initial_Secure to indicate the security context state.
    //@siclog "Step 16" siclog@
    //The UE transmits an ULInformationTransfer message and REGISTRATION COMPLETE message without including an SOR transparent container
    f_NR5GC_Registration_Complete(nr_Cell11);
    
    //@siclog "Step 17" siclog@
    //Check: Does the UE transmit an UL NAS TRANSPORT message with "Request type" set to "initial emergency request", and a PDU SESSION ESTABLISHMENT REQUEST message on NR Cell 13
    //@siclog "Step 17Aa1" siclog@Exception
    // The generic procedure for UE-requested PDU session establishmentspecified in subclause 4.5A.2,
    //takes place performing establishment of UE-requested PDU session(s) with ExpectedNumberOfNewPDUSessions = pc_noOf_PDUsSameConnection.
    
    if(not pc_noOf_PDUsSameConnection > 0)
    {
      v_InitialRxd := true;
    }
    
    // Clear store of msgs received in parallel
    f_NR5GC_MsgInDefault_ResetStoreOfNASMessageReq(); // @sic R5s220174 sic@
    // Then activate store for any new msgs received - this works for Modification Reqs too
    v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(nr_Cell11));

    while (not v_InitialRxd or not v_EmgRxd) {
      if (f_NR5GC_MsgInDefault_GetNumOfNASMessageReq() > 0) {
        v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOneNASMessageReq(1); 
      } else {
        SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell11, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;
        f_NR5GC_MsgInDefault_StoreNASMessageReq (v_ReceivedAsp);
      }
      if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_Nas_Transport.requestType, cr_NG_Request_Type('001'B))) { // initial
        v_InitialRxd := true;
        f_NR5GC_PDUSessionEstablishment (nr_Cell11, 1, v_SRB, 
                                         -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -,
                                         -, -, -, -, -, -, -, -, -, -, -, -, false, 1);
      } 
      else if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.ul_Nas_Transport.requestType, cr_NG_Request_Type('011'B))) 
      { // emergency
        v_EmgRxd := true;
      
        //@siclog "Step 18-22" siclog@//@sic R5-216165 sic@
        //Steps 9 to 13 of the procedure of "IMS Emergency call establishment in 5GC with IMS emergency registration" described in TS 38.508-1 [4] Table 4.9.11.2.2-1 are performed.
        f_NR5GC_EmergencyCallPDUSession(nr_Cell11, v_SRB, 1);//@sic R5s211479 sic@
        f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2], "Emergency Call Established");  // Emergency call now active @sic R%s220982, R5-225362 sic@    
      } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Wrong Request Type");
      }
      v_SRB := tsc_NR_RbId_SRB2;
      // Can remove the message that's been processed so that there's only 1 when calling f_NR5GC_PDUSessionEstablishment
      f_NR5GC_MsgInDefault_RemoveOneNASMessageReq();
      
    }

    //@siclog "Step 23" siclog@//@sic R5-216165 sic@
    //Generic procedure of "Test Procedure for IMS MO Emergency call release" described in TS 38.508-1 [4] Table 4.9.12A.2.2-1 are performed.
    f_NR5GC_EmergencyCallRelease(nr_Cell11);//@sic R5s211479 sic@
    // The above function won't finish until IMS messages have been sent and PDU session has been released @sic R5s220982 sic@
    deactivate(v_MyDefaultVar); //@sic R5-231404 sic@
    //@siclog "Step 24" siclog@//@sic R5-216165 sic@
    //The SS releases the RRC connection.
    v_MyDefaultVar := activate(a_NR5GC_Deregistration_REQ(nr_Cell11)); //@sic R5-231404 sic@
    f_NR_RRCRelease(nr_Cell11);
    deactivate(v_MyDefaultVar); //@sic R5-231404 sic@
    }  






